

# Introduction to Graphics Using SAS/GRAPH® Software

Mike Kalt  
Education Division  
SAS Institute



## Today's Menu

Section 1: Course Logistics and Overview
Section 2: Producing Scatter and Line Plots
Section 3: Producing Bar Charts
Section 4: Customizing the Appearance of SAS/GRAPH Output
Section 5: Generating Graphs for Use in Other Applications
Section 6: Conclusion



2

## Section 1: Course Logistics and Overview

### 1.1 Some Logistical Issues

### 1.2 Types of Output Produced by SAS/GRAPH

### 1.3 Overview of "Classic" SAS/GRAPH

### 1.4 SAS/GRAPH and the Output Delivery System (ODS)

3

## What This Presentation Is (and Isn't)

This is an **introduction** to creating basic graphics with SAS/GRAPH.

- The **SAS/GRAPH I: Essentials** introductory course from SAS Education runs 3 full days.
- This seminar runs 50 minutes. Do the math ☺
- You will learn:
  - How to create basic plots and charts
  - How to modify (somewhat) the appearance of graphs
  - How to export graphs to other applications

4

## Enterprise Guide vs. SAS/GRAPH

Enterprise Guide generates SAS/GRAPH code "behind the scenes".

- Enterprise Guide:
  - is easier to use
  - lets you design graphs using a point and click interface, and save or modify the generated code.
- SAS/GRAPH:
  - provides more powerful customization features
  - can produce some types of graphs that Enterprise Guide can't.

This seminar will show you how to write SAS/GRAPH programs. You can use what you learn here to write your own programs or to modify code generated by Enterprise Guide.

5

## Section 1: Course Logistics and Overview

### 1.1 Some Logistical Issues

### 1.2 Types of Output Produced by SAS/GRAPH

### 1.3 Overview of "Classic" SAS/GRAPH


### 1.4 SAS/GRAPH and the Output Delivery System (ODS)

6

## Two Types of SAS/GRAPH Output

SAS/GRAPH programs produce graphics using two very distinct systems. The two systems are:

- Original or "Classic" SAS/GRAPH ("G" procedures)
- Statistical Graphics ("SG" procedures).



This seminar primarily focuses on the original SAS/GRAPH procedures GCHART and GPLOT.

7

## Section 1: Course Logistics and Overview

1.1 Some Logistical Issues

1.2 Types of Output Produced by SAS/GRAPH

1.3 Overview of "Classic" SAS/GRAPH

1.4 SAS/GRAPH and the Output Delivery System (ODS)

8


## "Classic" SAS/GRAPH Procedures

"Classic" SAS/GRAPH procedures can be used to create the following types of graphs:

- two-dimensional scatter plots and line plots, including many types of statistical plots (GPLOT)
- bar, block, pie, and star charts (GCHART)
- three-dimensional scatter and surface plots (G3D)
- contour plots (GCONTOUR)
- maps (GMAP)
- text slides (GSLIDE)
- custom graphs (Annotate Facility)


9

## PROC GPLOT: Scatter and Line Plots




10

## PROC GCHART: Bar Charts




11

## PROC GCHART: Pie Charts




12

## PROC GMAP: Maps




13

## PROC G3D: 3-D Surface Plots/Scatter Plots




14

## PROC GCONTOUR: Contour Plots




15

## PROC GTILE: Tile Charts




16

## PROC GKPI: KPI Charts for Dashboards



17

## PROC GREPLAY: Multiple Graphs Per Page



18

## ANNOTATE Facility: Add Annotation to Graphs



19

## SAS/GRAPH Program Structure and Syntax

SAS/GRAPH procedure code uses the general syntax framework of other SAS procedures.

```
GOPTIONS graphics-specific-options;
<SAS/GRAPH global statements>
PROC graphics-procedure-name DATA=libref.data set;
    other procedure statements / <statement options>;
RUN;
```

20

## Sample SAS/GRAPH Program


The following program produces a horizontal (HBAR) chart showing a frequency count of the values of the variable `employee_country`:

```
title 'HBAR Bar Chart of Employees';
proc gchart data=employees;
    hbar employee_country;
run;
```

Most SAS/GRAPH procedures have a statement that invokes the procedure (PROC GCHART), and another "action" statement that requests a specific type of graph (HBAR).

21

## PROC CGHART: Example



22


## Modified SAS/GRAPH Program

Modify the previous program to produce a three-dimensional vertical bar chart:

```
title '3D VBAR Bar Chart of Employees';
proc gchart data=employees;
    vbar3d employee_country;
run;
```

23

## SAS/GRAPH Program Output



24

## Section 1: Course Logistics and Overview

1.1 Some Logistical Issues

1.2 Types of Output Produced by SAS/GRAPH


1.3 Overview of "Classic" SAS/GRAPH

1.4 SAS/GRAPH and the Output Delivery System (ODS)

25

## Output Delivery System

ODS can create output in different formats. Each ODS destination statement creates output for a specific type of viewer.



26

## Basic ODS Syntax

**ODS destination <options>;**  
*SAS/GRAPH (and/or other procedure) code*  
*to create a report*  
**ODS destination CLOSE;**

For example, to create a PDF file:

```
ods pdf file='sesug.pdf';
title 'VBAR Bar Chart of Employees';
proc gchart data=employees;
  vbar employee_country;
run;
ods pdf close;
```

27

## ODS LISTING Destination

By default, both text and graphics output go to the LISTING destination.

- For procedures that produce text output (e.g. PROC PRINT), the LISTING destination is the Output window.
- For SAS/GRAPH device-based procedures, the LISTING definition is the GRAPH1 window.

Since the LISTING definition is used by default, the following code produces a graph in the GRAPH1 window.


```
title 'VBAR Bar Chart of Employees';
proc gchart data=employees;
  vbar employee_country;
run;
```

No ODS statements are required

28

## Using ODS Styles

The default appearance of graphs and text reports produced by ODS is based on predefined styles. Styles dictate the default colors, fonts, and other attributes used for the output. Different default styles are used for each destination. The default style for the LISTING destination is shown here:



Styles are not used for graphs prior to the 9.2 release of SAS

29

## Using the ODS LISTING Statement

You can use the ODS LISTING statement with SAS/GRAPH to send output to the LISTING destination and use a different style:

```
ODS LISTING STYLE=style-name;  


SAS/GRAPH code to create a report  

RUN;
```

Since the LISTING destination is used by default, you only need to supply an ODS LISTING statement if you want to specify a style, or if you had previously closed the LISTING destination.


30

## Default Graph Style for LISTING Destination



31

## Using "ODS LISTING STYLE=BANKER;"



SAS provides over 50 predefined styles. You can also create your own custom styles.

32

## How to Change the Appearance of Graphs

- Specify the STYLE= option in an ODS statement to use predefined styles.
- Use GOPTIONS statements to specify attributes for graphics components.
- Specify options in SAS/GRAPH procedures or global statements to control appearance of specific attributes.

These methods will be discussed in Section 4, "Customizing the Appearance of SAS/GRAPH Output".

33

## Section 2: Producing Scatter and Line Plots

### 2.1 Creating Scatter Plots

### 2.2 Creating Line Plots

### 2.3 Creating Plots Containing Multiple Lines

### 2.4 Creating Other Types of Plots

34

## The GPLOT Procedure

General form of the GPLOT procedure to create scatter plots:

```
PROC GPLOT DATA = SAS-data-set;
  PLOT y-variable * x-variable / options;
RUN;
```


```
proc gplot data=budget;
  plot Yr2007 * Month;
run;
```

Vertical axis variable

Horizontal axis variable

35

## Scatter Plots



36

## SYMBOL Statements

You can specify alternate plotting symbols by using one or more SYMBOL statements. For example:

```
symbol1 value=dot cv=red height=12pt;
```

Plotting symbol

Symbol color

Height of symbol

Symbol height can be specified in the following units: CELLS (default), CM, IN, PT, or PCT. A cell is the height of one character in the text font used for the graph.


Colors will be discussed in more detail in Section 4.

37

## SYMBOL Statements

SYMBOL statements are global statements and can be placed outside PROC step

```
symbol1 value=dot cv=red
height=12pt;
proc gplot data=budget;
plot Yr2007 * Month;
```



38

## SYMBOL Statements

The following are some of the values that you can specify for the VALUE= option, and the corresponding symbol.

VALUE=	Plot Symbol	VALUE=	Plot Symbol
PLUS	+	TRIANGLE	△
X	×	DOT	●
STAR	*	CIRCLE	○
SQUARE	□	= (equals)	☆
DIAMOND	◇	: (colon)	*

VALUE==

VALUE=:

39

## Section 2: Producing Scatter and Line Plots

### 2.1 Creating Scatter Plots

### 2.2 Creating Line Plots

### 2.3 Creating Plots Containing Multiple Lines

### 2.4 Creating Other Types of Plots

40

## Line Plots


Use the interpolation (I=) option in the SYMBOL statement to create line plots.

I =	Purpose
NONE	specifies that no plot line be generated (default).
JOIN	connects the data points with a straight line.
SPLINE	connects the data points with a smoothed line.

41

## Using I=JOIN

```
symbol1 cv=green value=star i=join;
proc gplot data=budget;
plot Yr2007 * Month;
```




CV= option controls color of both symbols and lines

42



## Using I=SPLINE

```
symbol1 cv=green value=star i=spline;
proc gplot data=budget;
plot Yr2007 * Month;
```



43

## SYMBOL Statements

Options in the SYMBOL statement can control attributes of the plot line.

**SYMBOL***n* options;

Option	Purpose
<b>WIDTH</b> = <i>n</i> <b>W</b> = <i>n</i>	specifies the plot line thickness.
<b>LINE</b> = <i>line-type</i> <b>L</b> = <i>line-type</i>	specifies a plot line type (solid, dashed, dotted) value ranging from 1 to 46.
<b>CI</b> = <i>line-color</i>	specifies the color of the plot line only.
<b>COLOR</b> = <i>color</i> <b>C</b> = <i>color</i>	specifies the color for both plot symbol and plot line.

44

## SYMBOL Statements


The following is a sample of values that you can specify for the LINE= option and the corresponding line type.

1	_____
2	.....
3	-----
4	-----
5	-----
6	-----
7	-----
8	-----
9	-----
10	-----

45

## The SYMBOL Statement

```
symbol1 v=dot cv=blue h=2pct
i=join ci=red l=2 w=2;
proc gplot data=budget;
plot Yr2003 * Month;
```



46

## Section 2: Producing Scatter and Line Plots

2.1 Creating Scatter Plots

2.2 Creating Line Plots

2.3 Creating Plots Containing Multiple Lines


2.4 Creating Other Types of Plots

47

## Multiple Plot Requests

If you submit multiple PLOT requests in a single PLOT statement, each request produces a separate plot. The same SYMBOL statement is used for each plot.

```
symbol1 cv=green v=star i=join;
proc gplot data=budget;
plot Yr2006 * Month
Yr2007 * Month;
run;
```



48



## Plots Containing Multiple Lines

The following methods can be used to produce multiple plot lines within one set of coordinated axes:

- use the OVERLAY option in the PLOT statement to overlay two or more plot requests
- use a PLOT2 statement in addition to the PLOT statement to produce multiple plot lines with two vertical axes
- Use a third, classification variable to produce a separate plot line for each value of that variable.

49

## The OVERLAY Option

The OVERLAY option places all the plots that are generated by the PLOT statement on one set of axes.


```
plot Yr2007 * Month
     Yr2006 * Month / overlay legend;
```

Note the following when using the OVERLAY option:

- The horizontal axis variable should be the same for all plot requests.
- A legend is produced only if you include the LEGEND option.
- A separate SYMBOL definition is used for each plot.

50

## The OVERLAY Option



51

## The PLOT2 Statement

The PLOT2 statement produces a plot with a vertical axis on the right side of the graph. The syntax for the PLOT2 statement is identical to that for the PLOT statement.


Example:

```
plot Yr2007 * Month / legend;
plot2 Yr2006 * Month / legend;
```

To include a complete legend, the LEGEND option must be specified on **both** the PLOT and PLOT2 statements.

52

## The PLOT2 Statement



53

## Using a Third Variable

A third (classification) variable can be used to produce multiple plot lines, one for each value of the variable.

Example:

```
plot Profit * YYMM = Company;
```

- A separate plot line is produced for each value of **Company**.
- A PLOT statement of this form produces a legend automatically.
- A separate SYMBOL definition is used for each value of **Company**.


54

## Using a Third Variable

Separate SYMBOL statement used for each plot line

```
symbol1 c=blue v=triangle i=join;
symbol2 c=red v=square i=join;
symbol3 c=green v=diamond i=join;
proc gplot data=profit;
plot Profit * YMM = Company;
```

Separate plot line for each value of Company



55

## Section 2: Producing Scatter and Line Plots

2.1 Creating Scatter Plots

2.2 Creating Line Plots

2.3 Creating Plots Containing Multiple Lines

2.4 Creating Other Types of Plots


56

## Regression Interpolation: I=R

Specifying I=R draws a linear regression line through the points. CI= specifies the color of the regression line.

```
symbol1 v=dot i=r cv=red ci=red;
proc gplot data=budget;
plot Yr2007 * Month;
```

Additional regression interpolation options are available




57

## BOX Interpolation: I=BOX

I= BOX creates box plots, with the bottom and top edges of the box at the 25<sup>th</sup> and 75<sup>th</sup> percentile.


```
symbol1 co=blue i=box;
proc gplot data=profit;
plot Profit * year;
```

Additional box interpolation options are available




58

## Additional Regression and Box Plot Examples



59

## Some Additional Interpolation Methods



HILO Interpolation—joins highest and lowest Y values for a given X value

Needle Interpolation—draws vertical lines from points to horizontal axis

60

## Section 3: Producing Bar Charts

### 3.1 Creating Vertical and Horizontal Bar Charts

### 3.2 Creating Grouped and Subgrouped Bar Charts

61

## The GCHART Procedure

General form of the GCHART procedure to create vertical and horizontal bar charts:


```
PROC GCHART DATA = SAS-data-set;
  VBAR chart-variable(s) / options;
  VBAR3D chart-variable(s) / options;
  HBAR chart-variable(s) / options;
  HBAR3D chart-variable(s) / options;
RUN;
```

The **chart variable** (also known as the **midpoint variable**) is the variable that is used to determine the number of bars in the chart. By default, the length of the bar represents a frequency count for the values of the chart variable.

62

## Vertical Bar Charts

```
proc gchart data=orders;
  vbar Customer_Age_Group;
  vbar3d Customer_Age_Group;
```




63

## Horizontal Bar Charts

```
proc gchart data=orders;
  hbar Customer_Age_Group;
  hbar3d Customer_Age_Group;
```

Statistics are displayed by default in HBAR and HBAR3D charts



64

## Chart Variables

When the chart variable is a character variable, PROC GCHART draws a bar for each value of the variable.


When the chart variable is a numeric variable, PROC GCHART, by default, does the following:

- treats the variable's values as a continuous range and divides the range into intervals
- represents each interval with a bar
- labels the bar with the interval midpoint value

65

## If the Chart Variable is Numeric

In this example, the value of **Price** is divided into ranges and each bar represents a range. The bar label displays the midpoint of the range:



66

If the chart value is numeric and has a small number of values, the axis is still divided into ranges, and the output may not be appropriate:



The following options control the midpoints of bars in a chart:

68

## 69

## 70

## 71

## 72


## Specifying a Chart Statistic

The following options specify a statistic to be represented by the length of the bars in a chart:

Option	Purpose
<b>TYPE = statistic</b>	specifies what statistic the length of the bar represents. Possible statistics include FREQ (default), PERCENT, SUM, or MEAN. If you request SUM or MEAN, the SUMVAR= option is required.
<b>SUMVAR = numeric-variable</b>	specifies the variable to be used for sum or mean calculations.

73


## Specifying a Summary Statistic



74

## Specifying a Summary Statistic

```
proc gchart data=orders;
  hbar Customer_Age_Group /
  type=mean sumvar=Price;
```



Length of bars represents the mean value of **Price** for each value of **Customer\_Age\_Group**


Statistics displayed depend on the value of SUMVAR= and TYPE=

75

## Suppressing Statistics on HBAR Charts

```
proc gchart data=orders;
  hbar Customer_Age_Group /
  sumvar=Price
  type=sum
  nostats;
```

statistic represented by bars  
no statistics displayed on chart



76

## Section 3: Producing Bar Charts

### 3.1 Creating Vertical and Horizontal Bar Charts


### 3.2 Creating Grouped and Subgrouped Bar Charts

77

## Grouped and Subgrouped Bar Charts


The bars of a vertical and horizontal bar chart can be organized as follows:

- **Groups**, with a group of bars for each value of a group variable
- **Subgroups**, with each bar subdivided according to values of a subgroup variable



78

## Grouped Bar Chart




Separate set of bars for each value of **Customer\_Gender**

79

## Subgrouped Bar Chart

```
proc gchart data=orders;
  hbar Customer_Age_Group /
  sumvar=Price
  type=sum
  subgroup=Customer_Gender;
```



Bars are divided into segments based on value of **Customer Gender**

Each subgroup value is assigned a different pattern

80

## Section 4: Customizing the Appearance of SAS GRAPH Output

### 4.1 Overriding Style and Graph Attributes

### 4.2 Using the GOPTIONS statement

### 4.3 Customizing Graph Appearance

### 4.4 Customizing Axes

81


## Overriding Style and Graph Attributes

You can override default graph attributes using one or more of these methods:

- specify alternate ODS styles or create your own style
- use the GOPTIONS statement
- use options specified within the procedure, or on global statements such as TITLE, FOOTNOTE, AXIS, LEGEND, PATTERN, and SYMBOL statements

82

## Using an Alternate Style



83

## Section 4: Customizing the Appearance of SAS GRAPH Output

### 4.1 Overriding Style and Graph Attributes

### 4.2 Using the GOPTIONS statement

### 4.3 Customizing Graph Appearance

### 4.4 Customizing Axes

84

## The GOPTIONS Statement

The GOPTIONS statement specifies options that control various aspects of the SAS/GRAPH environment. The syntax is similar to the OPTIONS statement in Base SAS. For example:

```
goptions hsize=6 vsize=4 ; /* set height and width */
                        /* of graph */
```

To reset all GOPTIONS to their defaults, specify:

```
goptions reset=all;
```

85

## Using GOPTIONS to Control Color

Common GOPTIONS that impact color are shown below.

To...	Use This GOPTION
Specify the background color of the graph.	CBACK=
Select the default color for all text	CTEXT=

86

## CBACK= and CTEXT= Options

```
goptions cback=beige
        ctext=brown;
title 'Employees by Country';
proc gchart data=employees;
  vbar employee_country;
```



87

## Text Font and Height GOPTIONS

Common GOPTIONS that control text appearance are:

To...	Use This GOPTION
Set the default font for all text	FTEXT=
Specify the default height of all text	HTEXT=

If the font name contains a blank, the name must be enclosed in blanks. For example


```
goptions ftext='Arial Black';
```

88

## FTEXT= Option

The FTEXT= option affects the text in the graph output, including the title.

```
goptions ftext='Cumberland AMT/bold';
title 'Employees by Country';
title2 'Cumberland AMT bold font used for all text';
```



89

## Appearance GOPTIONS

GOPTIONS that control the size and appearance of the graphics area include:

To...	Use This GOPTION
Set the horizontal and vertical dimensions of the graph in inches	HSIZE= VSIZE=
Draw a border around the graphics area	BORDER
Specify an image file to display in a graph's background area	IBACK=


90



## Using the IBACK= Option

The default IBACK= behavior is for the image to be tiled in the graphics area.

```
options iback='c:\gears.gif';
```



91

## Section 4: Customizing the Appearance of SAS GRAPH Output

4.1 Overriding Style and Graph Attributes

4.2 Using the GOPTIONS statement

4.3 Customizing Graph Appearance

4.4 Customizing Axes

92

## SAS/GRAPH Colors

You can specify color names using any combination of these color-naming schemes, such as:

- RGB (red, green, blue) color values
- Gray scale color names
- SAS color names (defined in the SAS Registry)
- HLS (hue, lightness, saturation) values
- CMYK (cyan, magenta, yellow, black)
- HSV (hue, saturation, brightness).

93

## RGB Color Codes

Use the RGB color-naming scheme to specify a color in terms of its red, green, and blue components. Color names are of the form CXrrggbb, where

- CX indicates to use the RGB color specification
- rr is the red component
- gg is the green component
- bb is the blue component.

The components are given as hexadecimal numbers in the range 00 through FF. Most Web page (HTML) colors use RGB color values.

Example:

```
options cback=cx97b4a7;
```

94

## RGB Color Examples

Some selected RGB colors are shown below:

		Percent Blue					
Percent Red	Percent Green	00 = 0%	33 = 20%	66 = 40%	99 = 60%	CC = 80%	FF = 100%
CC = 80%	00 = 0%	cxCC0000	cxCC0033	cxCC0066	cxCC0099	cxCC00CC	cxCC00FF
	33 = 20%	cxCC3300	cxCC3333	cxCC3366	cxCC3399	cxCC33CC	cxCC33FF
	66 = 40%	cxCC6600	cxCC6633	cxCC6666	cxCC6699	cxCC66CC	cxCC66FF
	99 = 60%	cxCC9900	cxCC9933	cxCC9966	cxCC9999	cxCC99CC	cxCC99FF
	CC = 80%	cxCCCC00	cxCCCC33	cxCCCC66	cxCCCC99	cxCCCCCC	cxCCCCFF
FF = 100%		cxCCFF00	cxCCFF33	cxCCFF66	cxCCFF99	cxCCFFCC	cxCCFFFF

cxCCFF66 is:  
 • 80% red  
 • 100% green  
 • 40% blue

95

## Gray-Scale Color Codes

Use gray-scale codes to specify colors in terms of gray components. Gray-scale color names are of the form GRAYxx, where xx is the lightness of the gray.

The lightness component is given as a hexadecimal number in the range 00 through FF.

Color	Gray Scale Name
white (100% lightness)	GRAYFF
black (0% lightness)	GRAY00
60% lightness	GRAY99

Example:

```
options cback=grayaa;
```

96

## SAS Color Names

SAS provides a set of color names that you can use to specify colors.

Example:

```
options cback=FloralWhite
       ctext=blue;
```

97

## SAS Colors

Some selected SAS colors are shown below:

SAS Registry Name	RGB Name	Color It.
AliceBlue	CXF8F8FF	
AntiqueWhite	CXFAEB07	
Aqua	CX00FFFF	
Aquamarine	CX7FFD04	
Azure	CXF0FFFF	
Beige	CXF5F3DC	
Bisque	CXFFE4C4	
Black	CX000000	
BlanchedAlmond	CXFFEBCD	
Blue	CX0000FF	
BlueViolet	CX8A2BE2	
BR	CXA52A2A	
Brown	CXA52A2A	
Burlywood	CXD8BFD8	
FireBrick	CXB22222	
FloralWhite	CXFFFAF0	
ForestGreen	CX228B22	
Fuchsia	CXF000FF	

See the SAS Help System for more information on SAS color names.

98

## TITLE and FOOTNOTE Statements

General form of TITLE and FOOTNOTE statements:

```
TITLEn options 'text';
FOOTNOTEn options 'text';
```

Values for *n* range from 1 to 10. If *n* is omitted, it is assumed to be 1.

For SAS/GRAPH output, TITLE and FOOTNOTE statements can contain options that control attributes such as size, font, and color.

99

## TITLE and FOOTNOTE Statement Options

To...	Use This Option
Control font face	FONT= or F= title f='Albany AMT' 'Text';
Control color	COLOR= or C= title c=blue 'Text';
Control text height	HEIGHT= or H= title h=14pt 'Text';

You can control multiple attributes. For example:


```
title f='Arial' c=red h=3cm 'Text';
```

100

## Controlling the Appearance of Text

```
options ftext='Courier';
title f='Times New Roman' h=24pt c=brown
      'Employee Count';
footnote f='Arial Black' h=14pt c=gray33
        '2008 Data';
proc gchart data=employees;
  hbar employee_country /
  nostats;
```

GOPTIONS FTEXT= specifies the font for all text, but is overridden in TITLE and FOOTNOTE statements.




101

## Controlling the Width of Bars

Use the WIDTH= (or W=) option on the chart statement to control the width of bars:

```
proc gchart data=employees;
  where employee_country in
    ('AU' 'BE' 'DE');
  vbar employee_country /
  width=15;
```


Width is specified in units of character cells



102

## Bar Patterns

Each bar is filled with a pattern. For simple and grouped bar charts, the bars are filled with the default fill (SOLID) and the first data color specified in the style template



103


## Distributing Pattern Definitions

The PATTERNID= option distributes patterns in a bar chart:

Option	Purpose
<b>PATTERNID = MIDPOINT</b>	assigns a different PATTERN definition to each value of the chart (midpoint) variable.
<b>PATTERNID = GROUP</b>	assigns a different PATTERN definition to each value of the group variable.


104

## PATTERNID=MIDPOINT



105

## PATTERNID=GROUP



106

## The PATTERN Statement

You can assign patterns explicitly using a PATTERN statement. General form of the PATTERN statement:

**PATTERNn options;**

The following options can be specified:

Option	Purpose
<b>COLOR=   C=</b>	specifies the color of the fill pattern.
<b>VALUE=   V=</b>	specifies the fill pattern. The default fill is solid.

Example:


```
pattern1 c=red v=solid;
pattern2 c=blue v=empty;
```

107

## Using PATTERN Statements

The PATTERN statements below override the default patterns dictated by the style.

```
pattern1 c=lightblue v=s;
pattern2 c=lightgreen v=s;
pattern3 c=lightred v=s;
pattern4 c=lightorange v=s;
vbar employee_country /
patternid=midpoint;
```



108

## Section 4: Customizing the Appearance of SAS GRAPH Output

### 4.1 Overriding Style and Graph Attributes

### 4.2 Using the GOPTIONS statement

### 4.3 Customizing Graph Appearance

### 4.4 Customizing Axes

109


## Controlling Tick Marks in a Plot

You can use the following PLOT/PLOT2 statement options in the GPLOT procedure to control major and minor tick marks:

Option	Purpose
<b>HAXIS</b> = <i>value-list</i> <b>VAXIS</b> = <i>value-list</i>	specifies values for major tick marks on the horizontal axis and vertical axes
<b>HMINOR</b> = <i>number</i> <b>VMINOR</b> = <i>number</i>	specifies the number of minor tick marks to draw between each major tick mark on the horizontal and vertical axes.

110

## Controlling Tick Marks in a Plot



111

## Controlling Tick Marks in a Chart


You can use the following action statement options in the GCHART procedure to control major and minor tick marks:

Option	Purpose
<b>AXIS</b> = <i>value-list</i>	specifies values that correspond to major tick marks on the response axis.
<b>MINOR</b> = <i>number</i>	specifies the number of minor tick marks drawn between major tick marks on the response axis.

✍ The **response** axis displays the scale of values for the chart statistic.

112

## Controlling Tick Marks in a Chart



113


## Controlling the Axis Frame

You can use the following statement options in the GPLOT and GCHART procedures to control the appearance of the axis frame:

Option	Purpose
<b>CFRAME</b> = <i>color</i>	specifies a background color for the frame drawn around the axis area.
<b>NOFRAME</b>	specifies that no frame be drawn around the axis area.

114

## Controlling the Axis Frame



115


## Adding Reference Lines to a Plot

You can use the following PLOT/PLOT2 statement options in the GPROT procedure to add reference lines:

Option	Purpose
HREF = value-list	generates vertical lines that intersect the horizontal axis at the specified values.
VREF = value-list	generates horizontal lines that intersect the vertical axis at the specified values.
AUTOHREF	generates vertical lines that intersect the horizontal axis at major tick marks.
AUTOVREF	generates horizontal lines that intersect the vertical axis at major tick marks.

116

## Adding Reference Lines to a Plot



117


## Adding Reference Lines to a Chart

You can use the following chart statement options in the GCHART procedure to add reference lines:

Option	Purpose
REF = value-list	generates lines that intersect the response axis at the specified values.
AUTOREF	generates lines that intersect the response axis at major tick marks.
CLIPREF	clips the reference lines at the bars to produce the effect of the lines appearing behind the bars.

118


## Adding Reference Lines to a Chart



119


## AXIS and LEGEND Statements (Overview)

You can achieve much greater control over the appearance of your graphs by using AXIS and LEGEND statements.



120

## Plot Using AXIS and LEGEND Statements



121

## Plot Using AXIS and LEGEND Statements

```
options reset=all cback=verylightbrown;
symbol1 c=darkblue v=dot i=join w=3 h=2pct;
symbol2 c=darkred v='5F'x font='Monotype Sorts' i=join w=3 h=2pct;

axis1 w=4 label=(h=16pt f='Arial/bo' a=-90 r=90 'BUDGETED AMOUNT')
value=(h=12pt f='Arial/bo')
order=1000000 to 4000000 by 500000
major=(h=2 pct w=2) minor=(n=4 h=1 pct) length=80 pct
reflabel=(c=gray99 f='Arial/bo' h=10pt j=center 'Average');

axis2 value=(h=12pt a=45 f='Arial/bo' t=1 'JAN' t=2 'FEB' t=3 'MAR'
t=4 'APR' t=5 c=red 'MAY' t=6 'JUN' t=7 'JUL' t=8 'AUG'
t=9 c=red 'SEP' t=10 'OCT' t=11 'NOV' t=12 'DEC')
label=none minor=none major=(h=2 pct w=2) offset=(5,5) pct;

legend1 position=(top center inside) frame
cframe=verylightbrown cblock=black
label=(f='Arial/bo/it' h=12pt ' ')
value=(f='Arial/bo/it' h=12pt t=1 '2006' t=2 '2007');

proc gplot data=budget;
plot Yr2006 * Month Yr2007*month/ overlay legend vaxis=axis1
haxis=axis2 vref=2700000 grid legend=legend1;
title h=28pt 'Monthly Budget by Year';
run;
```

122

## Section 5: Generating Graphs for Use in Other Applications

### 5.1 Creating Image and Document Files

123

## Image and Document Files

In many cases, you will want to create output files in one of the following standard formats:

- Image Files (contain a graph only)
  - GIF
  - JPEG
  - Windows Metafile (WMF or EMF)
- Document files (can contain images and text)
  - HTML
  - PDF
  - RTF (Rich Text Format)

124

## Methods to Create Image and Document Files


You can use the following methods to create image and document files from SAS/GRAPH.

Method	Type of Output Created
Manually save an image file from the GRAPH1 window.	BMP, JPG, TIF, PNG, GIF, WMF, EMF, + more
Use ODS to create a document file	HTML, RTF, or PDF documents that include images and/or tables and text.

125

## Exporting Image Files from the GRAPH1 Window


After a graph is displayed in the GRAPH1 window, you can create an image file by selecting **File** ⇒ **Export as Image...** and then selecting the appropriate file type from the **save as type** box.



126

## Exporting Image Files from the GRAPH1 Window

The image file types are listed in the **Save as type** box.



127

## Creating HTML Files with ODS

The ODS HTML statement opens, closes, or manages the HTML destination.

To create an HTML file using ODS, submit the following:

```
ODS HTML PATH='directory-name'
      BODY='HTML-file-name';
SAS/GRAPH statements to produce graph
.
.
run;
ODS HTML CLOSE;
```

The ODS CLOSE statement must come after the RUN statement.

128


## Creating an HTML File with ODS

```
ods html path='c:\' body='sesug.html' style=analysis;
title 'Using ODS';
proc gchart data=employees;
vbar employee_country / patternid=midpoint;
run;
ods html close;
```

- The PATH= option specifies the directory into which both the HTML and image files will be placed.
- The BODY= option specifies the name of HTML file to create.

129

## HTML Output Using ANALYSIS Style



sesug.html

130


## Creating a PDF File

The FILE= option specifies the name and location of the PDF output file.

```
ods pdf file='c:\sesug.pdf';
title 'Using ODS';
proc gchart data=employees;
vbar employee_country / patternid=midpoint;
run;
ods pdf close;
```

131

## PDF Output



sesug.pdf

132



## Creating an RTF File


The FILE= option specifies the name and location of the RTF output file.

```
ods rtf file='c:\sesug.rtf';  
title 'Using ODS';  
proc gchart data=employees;  
  vbar employee_country / patternid=midpoint;  
run;  
ods rtf close;
```

133

## RTF Output


### Using ODS



sesug.rtf

134

## Graph After Editing in Microsoft Word



135

## Exporting Graphs to Microsoft Office

When exporting graphs to Microsoft Office applications the following methods are recommended:

- To insert a graph into an Excel spreadsheet, PowerPoint presentation, or existing Word document, create an EMF (Enhanced Windows Metafile) image file using the **File→Export as Image** pulldown in the GRAPH1 window. Use **Insert→Picture** in Excel, Powerpoint, or Word.
- To create a new Word document containing a graph, create an RTF file using ODS, and open the file in Word.

136

## Recommended Technical Documents

Detailed information on exporting SAS/GRAPH output to other environments can be found in the following SAS Technical Documents:

- TS-674, "An Introduction to Exporting SAS/GRAPH Output to Microsoft Office", at <http://support.sas.com/techsup/technote/ts674/ts674.html>
- TS-659, "Exporting SAS/GRAPH Output to PDF Files", at <http://support.sas.com/techsup/technote/ts659/ts659.pdf>

137

## Section 6: Conclusion

### 6.1 Additional Information

138

## SAS/GRAPH Training

The following courses are available from the SAS Education Division:

- **SAS/GRAPH 1: Essentials** -- covers device-based graphics
- **Statistical Graphics with ODS** -- covers SG procedures as well as statistical graphics that can be produced by procedures in Base SAS and SAS/STAT.

See <http://support.sas.com/training> for more information on these courses.

139

## Web Sites

The following web sites contain additional information and samples for SAS/GRAPH:

- <http://support.sas.com/sassamples/graphgallery/> -- a gallery of samples (including code) for both device-based and SG procedures.
- <http://www.robslink.com/SAS/Home.htm> -- hundreds of creative examples of creating complex graphs with SAS/GRAPH
- <https://support.sas.com/rnd/datavisualization/> -- lots of tips, programs, and downloads from the Data Visualization Division at SAS Institute.

140

**Thank you, and enjoy  
the conference! (what's left)**

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies. Copyright © 2010 SAS Institute Inc. Cary, NC, USA. All rights reserved.

